

Lean Audit for
*Scheduling with Uniform Tests: Exact Ratios,
Instance Optimality, and the Value of Information*

Václav Rozhoň

August 2026

Contents

1	Auditing the main statements in Lean	2
1.1	The common asymptotic meaning	3
1.2	Obligatory testing: exact competitive ratios	3
1.3	Obligatory-testing instance optimality	4
1.4	Why the obligatory instance assumptions are necessary	5
1.5	Blind execution	6
1.6	Blind optimization: the two worst-case curves	7
1.7	Blind-optimization instance optimality	8
1.8	Revealing optimization: exact competitive-ratio curves	9
1.9	Revealing-optimization instance optimality	10
1.10	Certified numerical summaries	11
1.11	Proof boundary and reproduction	11
	References	12

1 Auditing the main statements in Lean

The accompanying Lean 4 development [dMU21] uses mathlib [The20] and is pinned to version 4.30.0. This companion is organized as a reader-facing audit of the formal statements. Each item first states, in ordinary mathematical English, what one would like to verify and identifies its informal counterpart in the main paper. It then displays the decisive part of a public Lean signature and explains how to read it.

The excerpts suppress namespace prefixes, routine positivity hypotheses, and some local parameter definitions. Ellipses are expository: the exact export name is shown in every excerpt, and the source signature—not the excerpt—is authoritative about all arguments. In particular, an excerpt should not be read as evidence for a quantifier that it does not display. The concrete algorithms use finite uniform seeds for permutations, samples, and tie-breaking. The comparison and lower-bound theorems also have measure-theoretic corollaries for an arbitrary private probability space; these integrate the actual run cost directly.

paper result	principal Lean export	coverage	boundary/convention
deterministic OT ratio	<code>exact_ratio_main</code>	full	fixed inputs; completing strategy
randomized OT ratio	<code>paperGrowingPolicy_four_thirds_multiplicative; oblivious_iid_binary_lower_online_general_privateShuffle_actualOPT</code>	full	arbitrary private probability space; independent shuffle explicit
bounded OT instance comparison	<code>privateShuffle_paperGrowingPolicy_general_concrete_rate</code>	full	arbitrary private probability space and explicit placement average
OT instance-impossibility directions	<code>deterministic_balanced_cost_lower; exists_operational_quadratic_gap_general</code>	full	arbitrary private probability space in the randomized direction
bounded BE comparison and unbounded impossibility	<code>boundedUniform_blindPilot_general_concrete_rate; no_finite_ratio_at_quadratic_scale_operational_general</code>	full	arbitrary private probability space
BO curves and instance comparison	<code>deterministicCurve_operational_exact; exists_binary_mass_operational_ratio_tendsto_curve_general; generalExpectedCost_ge_instanceBenchmark_nsq</code>	full	arbitrary private probability space; placement average explicit
deterministic RO curve	<code>exact_ratio_main</code>	full	all positive finite caps
randomized RO curve	<code>exists_operational_binary_family_ratio_tendsto_curve_general</code>	full	arbitrary private probability space; fixed-input Yao selection
bounded RO instance comparison	<code>privateShuffle_compiled_le_generalCompetitor_concrete_rate</code>	full	arbitrary private probability space and explicit placement average

Here “full” refers to the main exact-ratio, instance-optimality, or impossibility conclusion and the quantifier convention in the last column, not to a line-for-line translation of the surrounding prose. In particular, the detailed sections below state when Lean uses a slower explicit vanishing rate than the sharper rate derived in the manuscript. For a general seed law μ , the Lean statements assume that the displayed real-valued expected costs are integrable. This is the usual well-formedness condition for a finite expected-cost comparison; a policy with infinite expected cost cannot violate any of the lower bounds.

The measure-theoretic bridge is in `GeneralRandomization.lean`. The paper-facing corollaries are in `GeneralRandomizedResults.lean`. Together they prove interchange of

a finite input average with integration over μ , apply the deterministic bound seedwise, and then select one fixed oblivious input outside the integral. Thus they do not use a finite-support approximation of the private seed.

1.1 The common asymptotic meaning

All competitive-ratio statements below use the following two predicates.

```
def SizeAsymptoticUpper (alg opt : forall n, Family n -> Real)
  (c : Real) : Prop :=
  forall eps, 0 < eps -> exists N, forall n, N <= n ->
    forall I, alg n I <= (c + eps) * opt n I

def FixedSizeLowerBound (cap : Cap) (ratio : Real) : Prop :=
  forall strategies : forall n, Online.Strategy n,
    forall eps, 0 < eps -> forall N,
      exists n, N <= n /\
        exists I : FixedInput cap n, exists fuel,
          I.defeats (strategies n) fuel (ratio - eps)
```

Here `I.defeats` means that the finite run has settled and either leaves a job incomplete or satisfies $(R - \varepsilon) \text{OPT}(I) \leq \text{ALG}(I)$. Thus the two predicates are exactly the upper and lower quantifier orders used for the size-asymptotic ratios in the paper; in particular, the lower input is fixed after the algorithm has been chosen.

1.2 Obligatory testing: exact competitive ratios

Theorem 1.1 (Lean-audited obligatory-testing ratios). *Let $z_* > 1$ be the unique solution of $z_* - 3 = \ln z_*$ and put*

$$R_* = 1 + \frac{2}{z_* - 1} = 1.57\dots$$

In obligatory testing, the optimal deterministic size-asymptotic competitive ratio is R_ . Against an oblivious adversary, the optimal randomized ratio is $4/3$. This is the formal counterpart of the main paper’s obligatory-testing theorem.*

For the deterministic statement, the public record contains both directions and an actual completing online strategy.

```
structure ObligatoryExactRatioConclusion (ratio : Real) : Prop where
  lower : LowerBound.FixedSizeLowerBound .infinite ratio
  upper :
    exists strategy : forall n, Online.Strategy n,
      UpperBound.CompletesByAnalysisFuel .infinite strategy /\
        SizeAsymptoticUpper
          (UpperBound.strategyCost .infinite strategy)
          (UpperBound.fixedOfflineCost .infinite) ratio

theorem exact_ratio_main :
  (forall {u : Real}, 0 < u ->
    FiniteExactRatioConclusion u (exactCurve u)) /\
    ObligatoryExactRatioConclusion RStar
```

The definitions in `SchedulingPaper/Constants.lean` identify `RStar` with R_* , while the second conjunct supplies the matching lower and upper bounds. The completion field rules out a certificate that only analyzes a partial transcript.

For the randomized statement, Lean exposes a finite upper error and a fixed binary input for the lower bound.

```

theorem unknownMultiset_operational_expectedCost_le_20378
  ... (p : Fin (k + rest) → Real) (hp : forall i, 0 ≤ p i) :
  uniformAverage
    (physicalSampledRunCost (k + rest) k d eta h_eta p) ≤
    4 / 3 * finiteObligatoryOPT p + 20378 * R ^ 7

theorem oblivious_iid_binary_lower_online_general_privateShuffle_actualOPT
  {n : Nat} (hn : 0 < n)
  (mu : Measure Omega) [IsProbabilityMeasure mu]
  (strategy : Omega → Online.Strategy n) (fuel : Nat)
  (hdone : forall seed input job, ... = .done) :
  exists input : BinaryInput n,
    (4 * n / (3 * n + 5)) *
      finiteObligatoryOPT (iidBinaryProcessingTime input) ≤
      generalExpectation mu fun seed =>
        uniformAverage fun order : Equiv.Perm (Fin n) =>
          Online.runCompletionCost .infinite ...

theorem paperGrowingPolicy_four_thirds_multiplicative
  (n : Nat) (hR : 12 ≤ fourthRoot n)
  (p : Fin (paperPilotSize n + paperRest n) → Real)
  (hp : forall i, 0 ≤ p i) :
  uniformAverage (physicalGrowingRunCost ... p) ≤
    (4 / 3 + paperWorstCaseError n) * finiteObligatoryOPT p

theorem paperWorstCaseError_tendsto_zero :
  Tendsto paperWorstCaseError atTop (nhds 0)

```

The public growing-policy theorem uses exactly the parameters printed for the main paper’s growing-cutoff algorithm: $d = \lfloor n^{1/4} \rfloor$, $k = \lfloor n^{3/4} \rfloor$, $B = 32 + n^{1/20}$, and $\eta = B/d$. The vanishing-error theorem therefore verifies the $4/3$ competitive ratio for $\mathcal{A}_n^*$ itself. Lean also retains the fixed-cutoff companion $\mathcal{A}_n^{(32)}$ and its sharper explicit $20378n^{7/4}$ additive bound. In the lower theorem the selected $\{0, 2\}$ input is outside both the seed and private-shuffle averages, and $4n/(3n + 5) \rightarrow 4/3$. The strategy is relabelled by the displayed uniform permutation, so the input remains oblivious to all private randomness.

1.3 Obligatory-testing instance optimality

Theorem 1.2 (Lean-audited obligatory instance comparison). *Fix $L < \infty$. There is one randomized nonanticipating algorithm, whose parameters depend on n but not on the input or on L , such that for every $p \in [0, L]^n$ and every completing announced algorithm $\mathcal{B}$ with integrable cost on an arbitrary private probability space,*

$$\mathbb{E} \text{ALG}_{\mathcal{A}^*}(p) \leq \mathbb{E} \text{ALG}_{\mathcal{B}}(p) + o_L(n^2).$$

The assignment of the job lengths in p to public labels is hidden by a private uniform permutation. This is the obligatory instance-optimal theorem in the main paper.

The final Lean comparison is between the physical learned run and the full private-shuffle average of an arbitrary family of completing competitors.

```

theorem privateShuffle_paperGrowingPolicy_general_concrete_rate
  (mu : Measure Omega) [IsProbabilityMeasure mu]
  ... {L : Real} (hL : 0 < L)

```

```

(p : Fin (k + r) -> Real)
(hp0 : forall job, 0 <= p job)
(hpL : forall job, p job <= L)
(competitor : Omega -> CompletePolicy p)
(htest : forall seed, FirstTouchesAreTests (competitor seed)) :
uniformAverage (physicalGrowingRunCost ... p) <=
  (k + r : Real) ^ 2 *
    (generalExpectation mu (fun seed =>
      uniformAverage (normalizedCost p (competitor seed))) +
      paperInstanceError L n)

theorem paperInstanceError_tendsto_zero (hL : 0 < L) :
  Tendsto (paperInstanceError L) atTop (nhds 0)

```

`FirstTouchesAreTests` is exactly the obligatory-testing restriction. The inner average is over hidden placements and the outer one over the competitor’s remaining seed; the second theorem turns the displayed normalized error into $o_L(1)$.

The corollary is proved by formally relabelling a completing policy. Lean checks that relabelling preserves completion and the first-touch testing restriction, and that its cost on a fixed placement σ equals the original cost on the composed placement $\sigma \circ \rho$. Postcomposition by fixed σ is a bijection of the finite permutation space, so averaging over the independent ρ produces exactly the inner uniform-placement average displayed above. Thus the private-shuffle bridge is part of the export rather than a prose-only step.

Both displayed exports have the identical left-hand term `physicalGrowingRunCost` with sample size $\lfloor n^{3/4} \rfloor$, grid size $\lfloor n^{1/4} \rfloor$, cutoff $32 + n^{1/20}$, and mesh B/d . The instance proof compares its fine-grid upper bound to an exact, grid-independent population optimum. It is then free to use a slower $m = \lfloor n^{1/32} \rfloor$ auxiliary grid solely for the adversarial lower bound on the competitor. Thus that scale is a proof device, not a second algorithm, and the “one algorithm” statement is now a literal fact of the Lean theorem signatures.

1.4 Why the obligatory instance assumptions are necessary

Lean also checks both limitations stated after the obligatory instance-optimal theorem in the main paper. Every completing deterministic policy has a fixed balanced zero–two input on which its literal cost is at least $9n^2/8$. Without a fixed input bound, every randomized policy with integrable cost on an arbitrary private probability space has a fixed input in $[0, n^2]^n$ on which it loses at least $n^2/8$ against the corresponding announced comparison schedule.

```

theorem deterministic_balanced_cost_lower
  (m : Nat) (strategy : Strategy (2 * m)) (fuel : Nat)
  (hcomplete : forall processing, ... -> CompletesAll processing strategy fuel) :
  exists processing,
    (forall job, processing job = 0 \/\ processing job = 2) /\
    (Finset.univ.filter fun job => processing job = 2).card = m /\
    9 * ((2 * m : Nat) : Real) ^ 2 / 8 <=
      runCompletionCost .infinite processing ...

theorem exists_operational_quadratic_gap_general
  {n : Nat} (hn : 7 <= n)
  (mu : Measure Omega) [IsProbabilityMeasure mu]
  (strategy : Omega -> Strategy n) (fuel : Nat)
  (hdoneHigh : ...) (hdoneZero : ...) :
  n ^ 2 / 8 <= averageCostOnAllHigh - announcedAllHighCost \/\
  exists label,
    n ^ 2 / 8 < averageCostWithOneZero label - announcedOneZeroUpper

```

The first input is frozen after the deterministic strategy is fixed. In the second theorem the exceptional zero label, when that branch is selected, is outside the private-seed average. These are the two operational quantifier orders required by the corresponding impossibility theorem in the main paper.

1.5 Blind execution

Theorem 1.3 (Lean-audited blind-execution statements). *For every fixed $L < \infty$, one randomized nonanticipating algorithm is instance-optimal on every $p \in [0, L]^n$ up to $o_L(n^2)$, even against an announced completing competitor. If no common upper bound on processing times is imposed, no randomized algorithm has a finite asymptotic competitive ratio. These are the bounded blind-execution theorem and its unbounded limitation from the main paper.*

The bounded comparison is a direct per-input inequality.

```
theorem boundedUniform_blindPilot_concrete_rate
  {n : Nat} ... {L : Real} (hL : 0 < L)
  (p : Fin n -> Real)
  (hp0 : forall job, 0 <= p job)
  (hpL : forall job, p job <= L)
  (competitor : CompletePolicy p) :
  uniformAverage (fun pilotOrder =>
    uniformAverage (blindPilotLearnedCost G pilot pilotOrder)) /
    (n : Real) ^ 2 <=
    uniformAverage (normalizedCost p competitor) +
    7830 * (L + 1) ^ 2 / concreteUnknownParameter n
```

The left side is the learned four-block algorithm; the right side is the private-shuffle cost of an arbitrary announced competitor. The checked parameter tends to infinity, so the last term tends to zero. The general corollary `boundedUniform_blindPilot_general_concrete_rate` performs the pointwise conditioning and measure-theoretic averaging in Lean itself.

For the unbounded part, Lean states the contradiction entirely in terms of literal completion costs.

```
theorem no_finite_ratio_at_quadratic_scale_operational_general
  {n : Nat} (hn : 0 < n)
  (mu : Measure Omega) [IsProbabilityMeasure mu]
  (strategy : Omega -> Strategy n)
  (hcomplete : forall seed, CompletesAllNonnegative (strategy seed))
  (R eta : Real) ...
  (hzeroCost : average zeroCost <= eta * n ^ 2)
  (hcompetitive : forall label,
    average (oneLongCost (n ^ 2) label) <=
      R * n ^ 2 + eta * n ^ 2) : False
```

Thus a low cost on the all-zero input and an R -competitive cost on every input with one length- n^2 job cannot hold simultaneously for large n .

1.6 Blind optimization: the two worst-case curves

Theorem 1.4 (Lean-audited blind-optimization curves). *For every fixed $u > 0$, the exact deterministic ratio is*

$$R_{\text{BO}}^{\text{det}}(u) = \begin{cases} 1, & u \leq 1, \\ u, & 1 \leq u \leq 2, \\ \frac{\sqrt{4u^3 - 4u + 1} - 1}{2(u - 1)}, & u \geq 2, \end{cases}$$

and the exact randomized ratio is

$$R_{\text{BO}}^{\text{rand}}(u) = \begin{cases} 1, & u \leq 1, \\ \frac{u^3}{u^2 + (u - 1)^3}, & 1 \leq u \leq u_{\text{BO}}, \\ \frac{1}{2} \left(1 + \sqrt{\frac{u^2 + u - 1}{u - 1}} \right), & u \geq u_{\text{BO}}, \end{cases}$$

where $u_{\text{BO}} > 1$ is the unique root of $u^3 - 2u^2 - u + 1 = 0$. This expands the two blind-optimization curves summarized in the main paper.

The Lean definitions contain these formulas literally.

```
def deterministicCurve (u : Real) : Real :=
  if u <= 1 then 1
  else if u <= 2 then u
  else (sqrt (4 * u ^ 3 - 4 * u + 1) - 1) / (2 * (u - 1))

def randomizedCurve (u : Real) : Real :=
  if u <= 1 then 1
  else if u <= randomizedTransition then
    u ^ 3 / (u ^ 2 + (u - 1) ^ 3)
  else
    (1 + sqrt ((u ^ 2 + u - 1) / (u - 1))) / 2
```

For the deterministic curve, one exported theorem gives both a completing upper strategy and a fixed input against every completing strategy, with linear remainders.

```
theorem deterministicCurve_operational_exact (hu : 0 < u) :
  (exists strategy : forall n, Online.Strategy n,
    (forall n, Online.CompletesAll u (strategy n)) /\
    forall n processing, admissible u processing ->
      runCost u processing (strategy n) n <=
        deterministicCurve u * offlineCost u processing +
        (1 + u) * n / 2) /\
  (forall n, 0 < n -> forall strategy,
    Online.CompletesAll u strategy ->
    exists processing, admissible u processing /\
      runCost u processing strategy n >=
        deterministicCurve u * offlineCost u processing -
        (2 * u ^ 2 + u) * n)
```

For randomization, the literal sample-based algorithm supplies the upper direction, and finite Yao selects a fixed binary input for the lower direction.

```
theorem operationalPilot_expectedCost_le_curve_normalized ... :
  uniformAverage (fun sampleOrder =>
```

```

conditionalExpectedPilotRunCost ... sampleOrder) <=
  randomizedCurve u * offlineCost u processing +
  pilotNormalizedError u n pilotSize * n ^ 2

theorem exists_binary_mass_operational_ratio_tendsto_curve_general
  (hu : 1 < u) :
  exists b in Set.Icc (0 : Real) 1,
  Tendsto (fun n => binaryFiniteYaoRatio n u b)
    atTop (nhds (randomizedCurve u)) /\
  forall n Omega (mu : Measure Omega), 0 < n ->
  forall strategy : Omega -> Strategy n,
  (forall seed, CompletesAll u (strategy seed)) ->
  exists input : BinaryInput n,
    binaryFiniteYaoRatio n u b * offlineCost u input <=
    generalExpectation mu fun seed => runCost u input (strategy seed) n

```

The theorem `pilotNormalizedError_tendsto_zero` verifies that any growing sublinear sample makes the upper error vanish.

1.7 Blind-optimization instance optimality

Theorem 1.5 (Blind-optimization instance statement). *For fixed $0 < u < \infty$, a sample-then-commit randomized algorithm satisfies, for every $p \in [0, u]^n$,*

$$\mathbb{E} \text{ALG}_{\mathcal{A}^*}(p) \leq \frac{n(n+1)}{2} \min\{u, 1 + \mu_p\} + o_u(n^2), \quad \mu_p = \frac{1}{n} \sum_i p_i.$$

Conversely, every completing algorithm with integrable cost on an arbitrary private probability space has expected cost at least

$$\frac{n^2}{2} \min\{u, 1 + \mu_p\} - nu\sqrt{2n}.$$

Hence both sides have the same leading value and Lean verifies the full blind-optimization instance comparison from the main paper.

Lean checks both the operational sample-then-commit construction and the arbitrary-policy lower benchmark:

```

theorem conditionalExpectedPilotRunCost_eq ... :
  conditionalExpectedPilotRunCost ... = conditionalExpectedPilotCost ...

theorem pilotPolicy_expectedCost_le ... :
  uniformAverage (fun sampleOrder =>
    conditionalExpectedPilotCost ... sampleOrder) <=
    n * (n + 1) / 2 * min u (1 + populationMean processing) +
    (1 + u) * pilotSize * n +
    n * (n + 1) / 2 * (u / sqrt pilotSize)

theorem generalExpectedCost_ge_instanceBenchmark_nsq
  (mu : Measure Omega) [IsProbabilityMeasure mu]
  ... (p : Fin n -> Real)
  (hp : forall job, p job in Set.Icc 0 u)
  (strategy : Omega -> Online.Strategy n)
  (hcompletes : forall seed placement,
    Online.Completes (placedProcessing p placement) (strategy seed)) :
  n ^ 2 / 2 * min u (1 + populationMean p) -
    n * u * sqrt (2 * n) <=
    uniformAverage fun seed =>

```



```

uniformAverage fun placement =>
  Online.runCost u (placedProcessing p placement)
    (strategy seed) n

theorem instanceBenchmark_normalizedError_tendsto_zero (u : Real) :
  Tendsto (fun n : Nat => u * sqrt (2 / (n : Real))) atTop (nhds 0)

```

The upper errors are $o_u(n^2)$ for any growing sublinear sample. The lower error is explicit; the third export proves that after division by n^2 it is $u\sqrt{2/n} = o_u(1)$. Its proof compiles a literal completing online strategy into a causal first-touch trace, applies a uniform-permutation first-moment bound at every position, and averages directly over the algorithm's private seed. Thus no conditioning step is left to prose.

1.8 Revealing optimization: exact competitive-ratio curves

Theorem 1.6 (Lean-audited revealing-optimization curves). *For every $u > 0$, the deterministic ratio is the six-piece function $R_{\text{RO}}^{\text{det}}(u)$ displayed in the main paper. The randomized ratio is*

$$R_{\text{RO}}^{\text{rand}}(u) = \begin{cases} 1, & 0 < u \leq 1, \\ \frac{u^3}{u^2 + (u-1)^3}, & 1 \leq u \leq u_2, \\ 1 + \frac{1}{2(\sqrt{u}-1)}, & u_2 \leq u \leq 25/4, \\ 4/3, & u \geq 25/4, \end{cases}$$

where $u_2 > 1$ is the unique root of $u^3 - 6u^2 + 5u - 1 = 0$. These are the revealing-optimization curves summarized in the main paper.

The deterministic curve is the first conjunct of the same top-level theorem used for the obligatory endpoint:

```

theorem exact_ratio_main :
  (forall {u : Real}, 0 < u ->
    FiniteExactRatioConclusion u (exactCurve u)) /\
    ObligatoryExactRatioConclusion RStar

def exactCurve (u : Real) : Real :=
  if u <= 1 then 1
  else if u <= uDiamond then u
  else if u <= uZero then rhoI u
  else if u <= goldenRatio + 2 then reciprocalBranch u
  else if u <= zStar then mixedFiniteCurve u
  else RStar

```

Here `FiniteExactRatioConclusion` has the same lower field, completing strategy, and `SizeAsymptoticUpper` field displayed at the start of this document. The branch functions and all five joins are proved in `SchedulingPaper/ExactCurve.lean`.

For the randomized curve, the scalar definition and the two attaining binary families are explicit:

```

def randomizedCurve (u : Real) : Real :=
  if u <= 1 then 1
  else if u <= transition then boundaryRatio u
  else if u <= 25 / 4 then interiorRatio u
  else 4 / 3

```

```

theorem binaryFamilies_le_curve (hu : 1 < u) :
  (forall tau, 1 <= tau -> tau <= u ->
    familyB u tau <= randomizedCurve u) /\
  (forall tau, 1 <= tau -> tau <= u - 1 ->
    familyA tau <= randomizedCurve u)

theorem binaryFamilies_attain_curve (hu : 1 < u) :
  (exists tau in Set.Icc 1 u,
    familyB u tau = randomizedCurve u) /\
  (exists tau in Set.Icc 1 (u - 1),
    familyA tau = randomizedCurve u)

```

The operational Yao theorem `exists_operational_binary_family_ratio_tendsto_curve_general` then selects one fixed labeled binary input outside an arbitrary private-seed integral. The theorem `randomizedRevealingCurveExact` is one public package containing this lower theorem, the announced operational upper theorem, the checked announced-to-unannounced transfer, and vanishing of the transfer error. The transfer itself is exported as `compiledExpectedNormalizedCost_le_curve_of_competitor`: given a announced policy satisfying the curve bound on every placement, it returns the corresponding direct bound for the literal unannounced sample-based learner.

1.9 Revealing-optimization instance optimality

Theorem 1.7 (Lean-audited revealing instance comparison). *Fix $0 < u < \infty$. There is one randomized nonanticipating algorithm such that for every $p \in [0, u]^n$ and every completing announced algorithm $\mathcal{B}$ with integrable cost on an arbitrary private probability space,*

$$\mathbb{E} \text{ALG}_{\mathcal{A}^*}(p) \leq \mathbb{E} \text{ALG}_{\mathcal{B}}(p) + o_u(n^2).$$

This is the revealing-optimization instance comparison in the main paper.

The final public theorem compares the literal learned strategy directly with the arbitrary competitor under the full private-shuffle average.

```

theorem privateShuffle_compiled_le_generalCompetitor_concrete_rate
  (mu : Measure Omega) [IsProbabilityMeasure mu]
  ... {u : Real} (hu : 0 < u)
  (p : Fin n -> Real)
  (hp0 : forall job, 0 <= p job)
  (hpu : forall job, p job <= u)
  (competitor : Omega -> CompletePolicy p) :
  compiledExpectedNormalizedCost grid sample u <=
    (generalExpectation mu fun seed =>
      uniformAverage (normalizedRawCost u p (competitor seed))) +
      concreteRevealingComparisonError u n

theorem concreteRevealingComparisonError_tendsto_zero (u : Real) :
  Tendsto (concreteRevealingComparisonError u) atTop (nhds 0)

```

The left side is an actual fixed-fuel transcript-only strategy averaged over independent sample and main permutations. The competitor may adapt among tests, processing, and raw execution and is given the input multiset. Its inner average ranges over hidden placements and its outer average over the remaining private seed; the second theorem supplies the required $o_u(1)$ normalized error.

The same checked relabelling construction used for obligatory testing proves this corollary. Lean additionally verifies that the revealing-model raw completion cost is

invariant under simultaneous relabelling of the policy and processing-time placement. The displayed export therefore packages the formerly implicit symmetrization step.

The concrete checked sample has size $\lfloor n/m^2 \rfloor = \Theta(n^{7/8})$, where $m = \lfloor n^{1/16} \rfloor$. This is now the parameter choice displayed in the introductory pseudocode as well.

1.10 Certified numerical summaries

The curve definitions remain exact, but the short decimal summaries in the introduction of the main paper are no longer justified only by external calculation. The file `SchedulingPaper/NumericalBounds.lean` exports, among others,

```
theorem RStar_decimal_bounds :
  (15705 : Real) / 10000 < RStar /\ RStar < 15706 / 10000

theorem uDiamond_decimal_bounds :
  (1866 : Real) / 1000 < uDiamond /\ uDiamond < 1867 / 1000

theorem uZero_decimal_bounds :
  (3147 : Real) / 1000 < uZero /\ uZero < 3148 / 1000

theorem RevealingOptimization.transition_decimal_bounds :
  (5048917 : Real) / 1000000 < transition /\
  transition < 5048918 / 1000000

theorem RevealingOptimization.globalMaximum_decimal_bounds :
  (1625 : Real) / 1000 < globalMaximum /\
  globalMaximum < 1626 / 1000
```

The algebraic intervals are discharged by exact polynomial arithmetic. For R_* , Lean brackets $\log q$ by a finite, rational `atanh` series, obtains explicit rational bounds on z_* , and then derives the displayed interval for R_* .

1.11 Proof boundary and reproduction

The root module `SchedulingPaper.lean` imports the developments used by all exports above. The checked project contains no `sorry`, `admit`, or project-defined `axiom`. It uses `mathlib`'s classical real analysis and noncomputable choices of exact real parameters; it proves mathematical properties of the algorithms but does not extract a floating-point implementation. The concrete randomized policies have finite uniform seeds, while the randomized lower and comparison corollaries quantify over arbitrary probability spaces with integrable real-valued run costs.

The coverage table at the start of this document is the authoritative summary. The arbitrary-seed private-shuffle corollaries for randomized obligatory testing and the bounded instance comparisons are explicit Lean exports. The obligatory 4/3 and bounded instance-optimal upper exports use the same `physicalGrowingRunCost` policy and the same concrete parameter family; the slower grid mentioned above is only a lower-bound proof device. Lean verifies the exact symbolic constants, branch transition equations, and the rational decimal brackets listed above. Longer printed digit strings remain readability aids unless a correspondingly narrow interval is explicitly exported.

The public development is available at

<https://github.com/vaclavrozhon/obligatory-testing>.

The accompanying source archive is

`obligatory-testing-lean-artifact.zip`.

It is the immutable reproduction target; its SHA-256 digest is recorded below. The file `lean-toolchain` pins Lean to `leanprover/lean4:v4.30.0`, and `lake-manifest.json` pins `mathlib` release `v4.30.0`. No archival DOI has been minted for this draft, so the archive digest, rather than the rolling branch, identifies the checked source exactly.

SHA-256:

`1a4d42f4ea22414cbf74f8de088769eaaf1ca947f48ca61bfeb7b0b62963a012`

From the unpacked archive root, the complete development is checked with

`lake build`

and each displayed export can be located by searching for its theorem name.

References

- [dMU21] Leonardo de Moura and Sebastian Ullrich. The Lean 4 theorem prover and programming language. In *Automated Deduction – CADE 28*, volume 12699 of *Lecture Notes in Computer Science*, pages 625–635. Springer, 2021. doi:10.1007/978-3-030-79876-5_37.
- [The20] The mathlib Community. The Lean mathematical library. In *Proceedings of the 9th ACM SIGPLAN International Conference on Certified Programs and Proofs (CPP 2020)*, pages 367–381. Association for Computing Machinery, 2020. doi:10.1145/3372885.3373824.